



www.oppadu.com

오피아두엑셀 무료 E-Book

## 1. 사용자 정의 폼 추가하기

예제파일에서 키보드 **Alt + F11** 키로 VBA 편집기를 실행합니다. **[삽입] - [사용자 정의 폼]**을 추가한 뒤, 아래와 같이 유저폼에 각 컨트롤을 추가합니다.

이름 : frmShtSelection  
Back Color : 흰색  
Caption : 시트를 선택하세요.

이름 : lstSheet  
MultiSelect : 2 - fmMultiSelectExtended

이름 : btnSubmit  
Back Color : 분홍색  
Caption : 시트병합

## 2. 사용자 정의 폼 초기화 명령문 작성 (UserForm\_Initialize)

사용자 정의 폼이 처음 실행되었을 시, 현재 **통합문서에 존재하는 워크시트를 리스트박스에 출력**하기 위한 명령문을 작성합니다. 유저폼안에 아래 명령문을 복사하여 붙여넣기 합니다.

```
Private Sub UserForm_Initialize ()

Dim WB As Workbook
Dim WS As Worksheet

Set WB = ThisWorkbook

For Each WS In WB.Worksheets
    Me.lstSheet.AddItem WS.Name
Next

End Sub
```



### 3. 시트 합치기 VBA 명령문 작성

#### 3-A. 변수 설정

아래와 같이 변수를 설정합니다.

- ✓ WB, WS : 사용중인 통합문서와 내용을 병합할 워크시트입니다.
- ✓ Rng : 각 시트에서 병합을 위해 붙여넣기 할 대상 범위입니다.
- ✓ newWS : 시트 병합을 위하여 새로 생성되는 워크시트입니다.
- ✓ i : For 문 / For Each 문을 반복을 위한 정수값입니다.
- ✓ j : 새로운 워크시트에 붙여넣기 할 시작셀 행번호입니다.
- ✓ endCol, endRow : 각 시트의 마지막 행/열 번호입니다.

```
Dim WB As Workbook
Dim WS As Worksheet: Dim newWS As Worksheet
Dim Rng As Range
Dim i As Long: Dim j As Long: j = 2
Dim endCol As Long: Dim endRow As Long

Set WB = ThisWorkbook
```

#### 3-B. 오류 처리

본격적인 시트 합치기 명령문을 실행하기 전, 오류 방지를 위해 아래 2 가지 오류를 처리합니다.

1. 리스트박스에서 아무런 항목도 선택되지 않은 채 사용자가 버튼을 클릭했을 경우
2. 기존에 '시트병합' 시트가 존재할 경우

```
'// isListBoxSelected 함수에 대한 자세한 설명은 아래 링크를 참고하세요.
'// https://oppadu.com/엑셀-사전/vba-리스트박스-값-선택여부/
If isListBoxSelected(Me.lstSheet) = False Then MsgBox "시트를 선택하세요.": Exit Sub

For Each WS In WB.Worksheets
    If WS.Name = "시트병합" Then MsgBox "시트병합 시트가 존재합니다. 시트병합 시트의 이름을 변경하시거나 삭제하신 뒤 다시 실행해주세요.": Exit Sub
Next
```



### 3-C. 새로운 워크시트 생성 및 머리글 입력

각 시트의 내용을 병합할 **새로운 워크시트를 생성**한 뒤, 머리글을 입력합니다.

```

Set newWS = WB.Worksheets.Add(before:=WB.Worksheets(1))
newWS.Name = "시트병합"
With newWS
    .Cells(1, 1) = "매장"
    .Cells(1, 2) = "날짜"
    .Cells(1, 3) = "이름"
    .Cells(1, 4) = "출근시간"
    .Cells(1, 5) = "퇴근시간"
End With

```


### 3-D. 시트 병합

리스트박스의 항목을 하나씩 돌려가며, 선택된 항목이 있는지 확인합니다. 만약 항목이 선택되어 있을 시, 해당 시트에서 마지막 행 및 마지막 열 번호를 받아 시트의 **마지막 셀 위치를 확인**합니다. 이후 복사/붙여넣기 할 범위를 지정한 뒤 새로 생성한 **'시트병합' 시트에 내용을 붙여넣기** 합니다.

```

For i = 0 To Me.lstSheet.ListCount - 1
    If Me.lstSheet.Selected(i) = True Then
        Set WS = WB.Worksheets(Me.lstSheet.List(i))
        With WS
            '/// 시트의 마지막 행/열 받아오기에 대한 자세한 설명은 아래 링크를 참고하세요.
            '/// https://oppadu.com/엑셀-사전/엑셀-vba-마지막-셀-찾기-마지막-행-찾기/
            endRow = .Cells(.Rows.Count, 1).End(xlUp).Row
            endCol = .Cells(1, .Columns.Count).End(xlToLeft).Column
            Set Rng = .Range(.Cells(2, 1), .Cells(endRow, endCol))
            Rng.Copy newWS.Cells(j, 1)
            j = j + Rng.Rows.Count
        End With
    End If
Next

```



### 3-E. 시트 합치기 완료 안내메세지 출력

시트 병합이 완료되었다는 **안내메세지를 출력**한 뒤, 유저폼을 종료합니다.

```
MsgBox "시트 병합이 완료되었습니다."  
Unload Me
```

### 3-F. 보조 함수/명령문 입력


보조 함수 및 명령문을 추가합니다. 해당 함수 및 명령문에 대한 자세한 설명은 오빠두 홈페이지에서 관련 포스트를 참고하세요.

```
Function isListBoxSelected(ListBox As MSForms.ListBox) As Boolean  
  
Dim i As Long  
  
For i = 0 To ListBox.ListCount - 1  
If ListBox.Selected(i) Then isListBoxSelected = True: Exit Function  
Next  
  
isListBoxSelected = False  
  
End Function
```

## 4. 유저폼 불러오기 단축키 설정

유저폼을 더욱 쉽게 불러올 수 있도록 단축키를 설정합니다. VBA 편집기에서 **[삽입] - [모듈]** 로 새로운 모듈을 추가한 뒤, 아래 명령문을 복사하여 붙여넣기 합니다.

```
Sub Merge_sheet()  
  
frmShtSelection.Show  
  
End Sub
```




www.oppadu.com

오빠두엑셀 무료 E-Book

이후 엑셀 시트로 돌아온 뒤, 상단의 [개발도구] - [매크로] 를 클릭하여 방금 생성한 'Merge\_sheet' 명령문을 선택한 뒤 [옵션]에서 **바로가기키 단축키를 설정**합니다.

만약 엑셀에서 개발도구가 보이지 않을 경우, 아래 관련 포스트를 참고하세요.

[엑셀 개발도구 활성화 방법 포스트 바로가기](#)



유저품을 불러오기 위한 바로가기키를 지정합니다.

## 5. 강의에 사용된 전체 명령문

### 5-A. 유저폼 명령문

```
Option Explicit

Private Sub btnSubmit_Click()

Dim WB As Workbook
Dim WS As Worksheet: Dim newWS As Worksheet
Dim Rng As Range
Dim i As Long: Dim j As Long: j = 2
Dim endCol As Long: Dim endRow As Long

Set WB = ThisWorkbook

'/// isListBoxSelected 함수에 대한 자세한 설명은 아래 링크를 참고하세요.
'/// https://oppadu.com/엑셀-사전/vba-리스트박스-값-선택여부/
If isListBoxSelected(Me.lstSheet) = False Then MsgBox "시트를 선택하세요.": Exit Sub
```



```
For Each WS In WB.Worksheets
    If WS.Name = "시트병합" Then MsgBox "시트병합 시트가 존재합니다. 시트병합 시트의 이름을 변경하시거나 삭제하신 뒤 다시 실행해주세요.": Exit Sub
Next

Set newWS = WB.Worksheets.Add(before:=WB.Worksheets(1))
newWS.Name = "시트병합"
With newWS
    .Cells(1, 1) = "매장"
    .Cells(1, 2) = "날짜"
    .Cells(1, 3) = "이름"
    .Cells(1, 4) = "출근시간"
    .Cells(1, 5) = "퇴근시간"
End With

For i = 0 To Me.lstSheet.ListCount - 1
    If Me.lstSheet.Selected(i) = True Then
        Set WS = WB.Worksheets(Me.lstSheet.List(i))
        With WS
            '// 시트의 마지막 행/열 받아오기에 대한 자세한 설명은 아래 링크를 참고하세요.
            '// https://oppadu.com/엑셀-사전/엑셀-vba-마지막-셀-찾기-마지막-행-찾기/
            endRow = .Cells(.Rows.Count, 1).End(xlUp).Row
            endCol = .Cells(1, .Columns.Count).End(xlToLeft).Column
            Set Rng = .Range(.Cells(2, 1), .Cells(endRow, endCol))
            Rng.Copy newWS.Cells(j, 1)
            j = j + Rng.Rows.Count
        End With
    End If
Next


MsgBox "시트 병합이 완료되었습니다."
Unload Me

End Sub

Private Sub UserForm_Initialize()

Dim WB As Workbook
Dim WS As Worksheet

Set WB = ThisWorkbook
```



```
For Each WS In WB.Worksheets
    Me.lstSheet.AddItem WS.Name
Next

End Sub

Function isListBoxSelected(ListBox As MSForms.ListBox) As Boolean

Dim i As Long

For i = 0 To ListBox.ListCount - 1
If ListBox.Selected(i) Then isListBoxSelected = True: Exit Function
Next

isListBoxSelected = False

End Function
```

5-B. 모듈 명령문

```
Option Explicit

Sub Merge_sheet ()

frmShtSelection.Show

End Sub
```